Про фотореализм
1. Общая информация об описаниях материала в POV-Ray
В POV-Ray для описания материала используется внутренний язык. У материала есть несколько основных разделов, таких как цвет поверхности, отражательные способности, неровности поверхности, внутренняя структура. Рассмотрим для примера сферу, имеющую следующий материал:

	pigment

{color rgb <1,1,1> }

finish

{diffuse 0.8 ambient 0.1 phong 0.8 phong_size 50 reflection 0.2 }  

normal

{bumps 0.1 scale 0.05}

Итак,  данный материал имеет три раздела, обозначающиеся ключевыми словами pigment, finish и normal. После ключевого слова в фигурных скобках записываются параметры раздела. 
Раздел pigment
Раздел «pigment» определяет цвет поверхности материала. Это цвет поверхности без учета освещения модели, без учета бликов и отражений.
В нашем материале в начале раздела pigment стоит ключевое слово color (это значит, что далее будет указан простой цвет, а не цветовая карта) и указан цвет. Кроме ключевого слова color может быть выбрана другая цветовая схема или инструкция для использования текстуры. Об этом будет написано чуть ниже, а пока рассмотрим использование слова color (я, чаще всего, использую только его, обычно этого хватает, чтобы делать довольно красивые картинки).
После ключевого слова color обычно записывается слово rgb, и в треугольных скобках указываются 3 составляющих цвета – красный, зеленый и синий. Обычно, цвет в RGB схеме в Windows кодируется целым числом от 0 до 255, но в POV-Ray используется другое представление. Каждая составляющая цвета – это число от нуля до 1. Единице в POV-Ray соответствует число 255 в Windows.
Таким образом, запись rgb <1,1,1> означает белый цвет. Если необходимо выбрать другой цвет, то лучше всего сделать следующее: в каком-нибудь графическом редакторе выбрать нужный цвет, записать 3 составляющих цвета в формате от 0 до 255 и разделить каждую составляющую на 255. После этого получившиеся десятичные составляющие можно записать в параметры цвета в POV-Ray.
Как уже было сказано, в данном разделе можно использовать и другие цветовые схемы (или процедурные текстуры, как еще они называются). Процедурные текстуры – это трехмерные математически заданные схемы размещения цветов. В POV-Ray их довольно много. Тут их описывать без толку, лучше всего почитать справку в самом POV-Ray. А я только приведу пример использования такой текстуры. Изменим немного материал.
	pigment

{crackle pigment_map{

[0 color rgb <1,0.5,0.5> ][1 color rgb <0.5,1,0.5>]}

   scale 0.2

}
finish

{diffuse 0.8 ambient 0.1 phong 0.8 phong_size 50 reflection 0.2 }  

normal

{bumps 0.1 scale 0.05}

Теперь в начале раздела pigment стоит слово crackle (это название процедурной текстуры) далее идет ключевое слово pigment_map. В некоторых процедурных текстурах цвет меняется плавно, по градиенту. Для того, чтобы описать этот градиент и нужно слово pigment_map. Описание градиента заключается в фигурные скобки, а каждый компонент градиента – в квадратные. После первой фигурной скобки стоит десятичное число, указывающее, в каком месте градиента располагается цвет. Цвет может быть задан с помощью тех же ключевых слов что и основной цвет.
Как бы так объяснить…цветовые схемы могут вкладываться друг в друга сколько угодно раз. Так вместо розового цвета (в pigment_map он записан первым) можно использовать схему checker (шахматное поле), тогда у нас получится следующая картинка.

	pigment

{crackle pigment_map{[0 checker color Red, color Red*0.5][1 color rgb <0.5,1,0.5>]}

   scale 0.2

}

finish

{diffuse 0.8 ambient 0.1 phong 0.8 phong_size 50 reflection 0.2 }  

normal

{bumps 0.1 scale 0.05}

После описания любой цветовой схемы, в разделе можно записывать различные модификаторы. Например, в нашем случае в конце стоит модификатор scale. Он изменяет масштаб текстуры с 1 (по-умолчанию) на 0.2. Кроме изменения масштаба можно использовать модификаторы translate (перемещение), rotate (вращение), turbulence (искажение). Модификаторы влияют на ту цветовую схему, после описания которой они идут. В нашем случае модификатор scale относится к цветовой схеме crackle.
Теперь что касается использования файловых текстур:

В настоящее время автоматически подгрузить текстуры из T-Flex в POV-Ray нельзя. Если необходимо использовать файловую текстуру, то необходимо вручную прописывать параметры в окне «Материал POV». Вот как должен выглядеть раздел pigment чтобы использовалась файловая текстура:
	pigment {

      image_map {jpeg "C:\im\New-Year.jpg" }scale 2 translate <-0.4,-0.5,0>

    }

finish

{diffuse 0.8 ambient 0.1 phong 0.8 phong_size 50 reflection 0.2 }

Сразу в начале раздела pigment должно стоять слово image_map, затем в фигурных скобках: сначала тип файла (POV-Ray принимает файлы типов gif | tga | iff | ppm | pgm | png | jpeg | tiff | sys), затем полный путь к файлу в кавычках. Нужно помнить, что POV-Ray русских букв не понимает, поэтому имена папок и файлов должны содержать только латинские символы.

При работе с текстурами есть существенные ограничения.

1) Если объект сложный, то трудно угадать, как именно T-Flex натянет текстуру на объект (а затем вершины этого объекта будут иметь в POV-Ray соответствующие текстурные координаты).  При этом обернуть текстуру вокруг объекта вообще нельзя, в T-Flex таких средств нет. Картинка всегда накладывается на объект с определенной стороны.
2) Что-то странное происходит при переносе масштаба и положения картинки из T-Flex в POV-Ray. Поэтому приходится вручную подбирать масштаб, положение и угол поворота картинки с помощью инструкций scale, translate и rotate.
В результате, я рекомендую использовать текстуры только на плоскостях. И только в том случае, если другого выхода нет.
В заключение о разделе pigment. Надо учитывать: Не факт, что результирующий цвет объекта на полученной картинке будет полностью совпадать с тем цветом, который был задан в разделе пигмент. Так как цвет объекта на картинке будет зависеть от таких параметров как освещенность объекта, отражательные характеристики поверхности, от того, что окружает объект и др. Поэтому, часто приходится немного изменять цвет в ту или другую сторону, чтобы на результирующей картинке цвет был таким как надо.

Раздел finish
Данный раздел предназначен для задания параметров того, как поверхность реагирует на свет. Реакция включает следующие эффекты:
1) Диффузное отражение. Это свет, который поверхность объекта рассеивает, придает ему цвет своей поверхности и отражает назад. Яркость диффузного отражения зависит от угла между нормалью поверхности и направлением падения света. Это основная реакция поверхности на свет. 
2) Блик. Твердые поверхности (или очень гладкие поверхности), например, металлы, часть света (напрямую от источника света) сразу отражают назад, прямо в «глаз» наблюдателя. Рассеивания света при этом почти не происходит. Поэтому на поверхности образуется яркое пятно, соответствующее цвету источника света.
3) Зеркальное отражение. Думаю, название говорит само за себя.

Конечно, это только основные эффекты вызванные реакцией поверхности тела на свет. Но, пока, достаточно и их.

Рассмотрим пример:

	pigment {  color rgb<1,0.4,0.4>  }

finish

{diffuse 0.8 ambient 0 phong 0.8 phong_size 50}

В разделе finish первым задается параметр diffuse. Данный параметр определяет диффузное отражение. Число 0.8, стоящее после слова diffuse означает, что 80% света падающего на поверхность рассеивается.
После этого определен параметр ambient. Это параметр задает яркость неосвещенных мест объекта. В нашем случае, с неосвещенной стороны сфера имеет черный цвет. Можно задать вопрос, зачем же нужен данный параметр, ведь если объект не освещен, он должен быть черным. Но рассмотрим нашу сцену: в реальности задняя часть сферы тоже была бы освещена – тем светом, который рассеяла и отразила поверхносить под сферой. Для симуляции этого эффекта параметр ambient можно задавать отличным от нуля, например 0.2. Важно, что данный параметр проявляется только при наличии в сцене источника рассеянного (всенаправленного) света. Что касается T-FLEX CAD, то яркость рассеянного источника задается в свойствах 3D-сцены. Но нужно учитывать, что данный параметр влияет на освещение всей сцены, так что, например, при увеличении яркости рассеянного света может потребоваться уменьшить интенсивность остальных источников.
Третий параметр, который мы задаем – phong. Это яркость блика на поверхности объекта. Число 0.8 означает, что в блик уходит 80% энергии света.
Далее идет параметр phong_size, равный 50. Данным параметр задает величину блика. Чем выше число, тем более острый получается блик. Чаще всего, величина блика в реальном мире определяется твердостью материала. Чем выше твердость, тем острее блик.

В нашем примере отсутствует зеркальное отражение. Добавим еще один параметр в раздел finish

	pigment {  color rgb<1,0.4,0.4>  }

finish

{diffuse 0.8 ambient 0 phong 0.8 phong_size 50 reflection 0.2}

Теперь сфера зеркально отражает свое окружение, так как в разделе finish появился параметр reflection, равный 0,2. 0,2 – коэффициент отражения. Если задать коэффициент равным 1, сфера станет зеркальной.
Раздел normal

Раздел normal испльзуется для симуляции неровностей поверхности материала. Это именно симуляция неровностей. Геометрия тела не изменяется, изменяется только направление нормалей поверхности тела (в зависимости от цвета используемой для этого текстуры). В предыдущий пример добавил раздел normal с некоторыми параметрами.

	pigment {  color rgb<1,0.4,0.4>  }

finish

{diffuse 0.8 ambient 0 phong 0.8 phong_size 50 reflection 0.2}
normal

{agate 0.1 scale 0.5}

Теперь сфера выглядит неровной, и все благодаря добавлению нового раздела. Раздел normal во многом напоминает раздел pigment, разница только в том, что нам нет необходимости определять конкретные цвета входящие в текстуру. В нашем примере, в начале также стоит имя agate - имя процедурной текстуры. За именем текстуры идет десятичное число определяющее глубину неровностей, а затем и параметр scale, определяющий масштаб неровностей.

Использование неровностей часто позволяет сделать изображение более реалистичным.
Теперь рассмотрим раздел о котором я ранее не упоминал.

Раздел interior

Данный раздел предназначен для описания внутренней структуры материала. Теоретически, внутренность объекта может быть заполнена структурой наподобие дыма или облаков. Но у меня ни разу этого нормально сделать не получалось. В результате я использую этот раздел только чтобы выбрать коэффициент преломления материала в том случае, если поверхность материала прозрачная. Для указания коэффициента преломления используется параметр ior.
Лучше сразу рассмотреть случай, когда материал прозрачный:
	pigment {   color rgb<1,0.4,0.4> filter 1   }

finish

{diffuse 0.8 ambient 0 phong 0.8 phong_size 50 reflection 0.2} 
interior{ior 1.0333}

Прозрачность поверхности задается параметром filter в разделе pigment. Данный параметр задается десятичным числом от 0 до 1. 1 – объект прозрачный, 0 – непрозрачный. Обратите внимания, что материал все равно имеет тот цвет, который задается ключевым словом color или текстурной картой.
В самом конце нашего материала стоит раздел interior. В этом разделе задан параметр ior, равный 1.0333.

Вот, пожалуй и все, про язык написания материалов в POV-Ray.

Теперь о T-Flex. Все разделы материала кроме раздела interior записываются в свойствах материала в окне «Параметры POV» в поле «Текстура: Описание». В этом поле описание материала пишется целиком, с указанием используемого раздела.

А вот раздел interior записывается в этом же окне, но уже в поле «Внутренность: Описание». Причем, само название раздела interior записывать не нужно, в поле пишутся сразу параметры этого раздела.

В поставку T-Flex входят 3 библиотеки материалов. Стандартная, камень и дерево. Обычно я использую только их, а если надо сделать что-то новое, модифицирую уже существующие материалы. При этом материал в библиотеке может быть задан в двух формах. Например: 
Первая форма:

T_Chrome_2A 

Как видите, напрямую параметры материала здесь не указываются. Вместо этого при рендеринге POV-Ray вместо слова T_Chrome_2A подставляет материал из библиотеки. В папке куда ставится POV-Ray есть папка includes, в ней куча файлов POV-Ray, содержащих готовые наборы параметров различных материалов. Их можно открывать и править с помощью POV-Ray, но лично я этим никогда не занимался, так как самостоятельно написать такой же качественный материал как, например, «гранит» из библиотеки POV-Ray довольно сложно.

Вторая форма:

pigment{color Red*0.9}

finish{diffuse 0.8 ambient 0 phong 0.4 phong_size 25 reflection 0.8}
В этой форме записи вам уже должно все быть понятно. Единственное пояснение: в разделе pigment после слова color стоит «Red*0.9». Тут опять используются библиотеки POV-Ray. Цвет умножается на 0.9, чтобы с сделать исходно ярко-красный цвет немного более темным.
Если в материале используются какие-либо библиотеки, для их указания используется поле «Включить файлы». Включаемые файлы в данном поле перечисляются через точку с запятой.
Теперь рассмотрим другие вопросы.

2. Выбор и настройка качества изображения

Создать достаточно удачное фотореалистичное изображение с первой же попытки удаётся редко. Обычно, требуется создать несколько тестовых фотореалистичных изображений, которые помогают скорректировать положение камеры,  яркость и положение источников света, а также проверить правильность анимации. После этого проводится финальная визуализация.

Но создание фотореалистичного изображения может занимать различное время, в зависимости от сложности сцены и параметров, определяющих качество изображения. Знание этих параметров с одной стороны, помогает избежать излишних затрат времени на проведение пробной визуализации, а с другой стороны, помогает добиться более высокого качества финального изображения.

Существуют различные параметры, позволяющие менять качество получаемого фотореалистичного изображения.

Плотность сетки. Данный параметр задаётся в статусе документа (команда «ST: Задать параметры документа»), и кроме фотореализма, влияет ещё и на качество отображения объектов в 3D окне. Чем выше данный параметр, тем дольше ведётся экспорт сцены в формат POV, тем больше оперативной памяти используется POV-Ray и тем дольше POV-Ray ведёт предварительную подготовку сцены перед визуализацией (Parsing). В связи с этим, при проведении предварительной визуализации плотность сетки желательно снизить, возможно, даже до минимума. При проведении финальной визуализации лучше установить максимальную плотность сетки.

Параметры источников света. При создании фотореалистичного изображение с использованием обычных источников света, тени объектов получаются очень чёткие, так как источники света имеют бесконечно малую величину. В реальности, это бывает очень редко, и тени, чаще всего, бывают сглаженными. Применение рассеянных источников света позволяет сделать тени более сглаженными и повысить качество и реалистичность изображения. В рассеянных источниках света вместо одного точечного источника света используются несколько сдвинутых друг относительно друга точечных источников. Чем больше они сдвинуты, тем менее чёткой будет тень. Чем больше количество точечных источников имеет рассеянный источник – тем выше размытость тени и тем больше времени нужно на визуализацию.

Обычный свет

Рассеянный свет


Рассеянный источник света в POV-Ray представляет собой множество точечных источников света.

	Эти источники света размещаются в виде прямоугольника, ориентированного некоторым образом относительно указанного центра. Количество источников света вдоль каждой из сторон прямоугольника может быть разное.

Чтобы созданный в T-FLEX CAD источник света стал рассеянным источником света в POV-Ray, в 

свойствах источника света в поле «Инструкции POV» необходимо записать следующее:
area_light <0.035, 0, 0>, <0, 0.035, 0.035>, 5, 5 adaptive 1 jitter
Здесь в треугольных скобках заданы координаты противоположных углов прямоугольника относительно исходной точки (точки, в которой расположен рассеянный источник света). «5, 5» - это количество источников света в каждом направлении. При этом общее количество точечных источников света составляет 5х5=25. «adaptive 1 jitter» - дополнительные параметры, включающие оптимизацию расчёта теней. 

В реальном мире, количество света, попадающего на поверхность предмета, уменьшается при удалении от источника света (это не касается солнечного света и направленных прожекторов). Эффект затухания света (Light Fading) может быть воспроизведен в POV-Ray, что в некоторых случаях может повысить качество фотореализма. Для того, чтобы включить затухание у источника света ему необходимо установить дополнительные параметры: fade_distance and fade_power. В таком случае в поле «Инструкции POV» может стоять, например:

area_light <0.035, 0, 0>, <0, 0.035, 0.035>, 5, 5 adaptive 1 jitter fade_distance 2 fade_power 3
В строке выше, параметр fade_distance определяет расстояние, на котором объекты освещены 100% яркостью, при увеличении расстояния яркость освещения падает. Параметр fade_power является целым числом от 1 до 3 и указывает скорость, с которой происходит падение интенсивности света с увеличением расстояния до источника света.
Хитрости при подготовке моделей к фотореализму:

В реальном мире большинство вещей не имеют абсолютно острых ребер. А вот 3D модель такие ребра имеет. В результате при создании фотореализма это бывает очень заметно. Поэтому, если это не занимает много времени, видные ребра модели можно скруглить небольшим радиусом, например 0.2 мм. В результате модель будет выглядеть более реалистично и красиво (на скругленных гранях могут появится легкие блики, которые лучше подчеркнут геометрию модели).
Сглаживание (Antialiasing). При обычной визуализации на границах объектов могут возникать явления ступенчатости, прерывистости тонких линий.  Сглаживание путём проведения дополнительных расчётов может уменьшить негативное влияние данных явлений.

Ступенчатость на границах

Сглаживание включено


Сглаживание основано на визуализации частей сцены с увеличенным разрешением. При этом визуализация сцены замедляется. Поэтому на этапе пробной визуализации включать сглаживание не следует. Но для финальной визуализации включить сглаживание желательно. 

Рассеянное освещение (Radiosity). При обычной визуализации учитывается прямое освещение, при котором освещены только те участки объектов, на которые непосредственно попадает свет от источника света. Однако в реальном мире, свет исходит не только от источников. Он ещё и отражается от объектов, освященных прямым светом. В POV-Ray есть возможность включить механизм расчёта рассеянного освещения, который, в некоторых случаях, помогает улучшить реалистичность изображения.

Обычное освещение

Рассеянное освещение


В связи с большим количеством дополнительных расчётов, использование механизма рассеянного освещения может привести к существенному замедлению визуализации. Поэтому использование рассеянного освещения при тестовой визуализации следует проводить только в низких разрешениях.

Чтобы включить механизм рассеянного освещения следует в поле «Включить строки» окна «Создание фотореалистичного изображения» записать следующее:

global_settings {

radiosity { count 500 minimum_reuse 0.018 brightness 0.8}}

Значение приведённых инструкций, а также дополнительные сведения, касающиеся механизма рассеянного освещения, следует искать в документации приложения POV-Ray.

Разрешение изображения. Данный параметр существенно влияет на время, затрачиваемое на визуализацию. При неизменном качестве изображения, скорость визуализации прямо пропорциональна площади результирующего изображения. При тестовой визуализации можно ограничиться небольшими разрешениями, например, 320*240.
Многопроцессорность
Новые версии POV-Ray (на январь 2011 года доступна версия 3.7.0RC3) поддерживают многопроцессорность. С точки зрения программирования, задачи расчета фотореализма распараллеливаются очень хорошо, поэтому при увеличении количества процессоров в 2 раза, почти (но не совсем) в 2 же раза уменьшается и время рендеринга. Поэтому, при создании фотореализма сложных сцен лучше выбирать компьютер, с многоядерным процессором.
